Diffusion Examples

William O. Bray
University of Maine

The purpose of this notebook is to illustrate graphically the solution to the heat equation on the real line. The intial value problem has the form

$u_t = \alpha^2 u_{xx}, \; x \in \mathbb{R}$

IC: $u(x, 0) = f(x)$

The Gaussian

In[29]:=

```
s[x_, t_] := Exp[-x^2 / (4 t)] / (4 Pi t)^(1/2)
```

The above defines the heat kernel. In what follows are several examples of solutions to the Cauchy problem for the heat equation. These illustrate ideas presented in class.

In[30]:=

```
Plot[{s[x, 1], s[x, 0.5], s[x, 0.1], s[x, 0.05]}, {x, -3, 3},
 PlotRange -> Full, PlotStyle → Thick, ImageSize → {500, 300}]
```

Out[30]=



Example 1

In[31]:=
```
f1[x_] := Piecewise[{{1, Abs[x] ≤ 1}, {0, Abs[x] > 1}}]
u1[x_, t_] := Evaluate[Integrate[f1[y] s[x - y, t], {y, -1, 1}]]
u1[x, t]
```

Out[33]=

$$\frac{1}{2}\left(-\mathrm{Erf}\left[\frac{-1+x}{2\sqrt{t}}\right]+\mathrm{Erf}\left[\frac{1+x}{2\sqrt{t}}\right]\right)$$

Notice that the above formula is the same as obtained in the text obtained by hand calculation and using the definition of the error function.

In[34]:=
```
Plot[{f1[x], u1[x, 0.0005],
  u1[x, 0.005], u1[x, 0.05], u1[x, 0.5]}, {x, -3, 3},
 PlotStyle → {Thick, {Dashed, Thick}, {Green}, {Blue, Thick}, Red},
 ImageSize → {500, 300}]
```

Out[34]=



A three dimensional rendition of the solution is given as follows.

In[35]:=
```
Plot3D[u1[x, t], {x, -4, 4}, {t, 0, 3},
 ImageSize → {500, 300}, AxesLabel → {x, t, u}]
```

Out[35]=



Example 2

In[37]:=
```
f3[x_] := Sign[x] Exp[-Abs[x]]
u3[x_, t_] := Evaluate[Integrate[f3[y] s[x - y, t],
   {y, -Infinity, Infinity}, Assumptions → {t > 0}]]
Simplify[u3[x, t]]
```

Out[39]=

$$-\frac{1}{2} e^{t-x} \left(-1 + \text{Erf}\left[\frac{2\,t - x}{2\sqrt{t}}\right] + e^{2x}\,\text{Erfc}\left[\frac{2\,t + x}{2\sqrt{t}}\right]\right)$$

In[40]:=

```
Plot[{f3[x], u3[x, 0.005], u3[x, 0.05], u3[x, 0.5]}, {x, -4, 4},
 PlotStyle → {Thick, {Dashed, Red}, {Green, Thick}, Blue},
 ImageSize → {500, 300}]
```

Out[40]=



The initial data above is an odd function. The following plot illustrates that our solution also gives the solution to an IBVP on the half-line with homogeneous Dirichlet boundary condition at $x = 0$. This foreshadows the method of reflection discussed in Chapter 4 of the text.

In[41]:=
```
Show[%, PlotRange → {{0, 4}, {0, 1}}]
```

Out[41]=



Example 3

In[42]:=
```
f4[x_] := x Exp[-Abs[x]]
u4[x_, t_] := Evaluate[Integrate[f4[y] s[x - y, t],
    {y, -Infinity, Infinity}, Assumptions → {t > 0}]]
Simplify[u4[x, t]]
```

Out[44]=

$$\frac{1}{2} e^{t-x} \left(-2 t + 2 e^{2 x} t + x + e^{2 x} x - e^{2 x} (2 t + x) \operatorname{Erf}\left[\frac{2 t + x}{2 \sqrt{t}}\right] \operatorname{Sign}[2 t + x]^2 + (2 t - x) \operatorname{Erf}\left[\sqrt{t} - \frac{x}{2 \sqrt{t}}\right] \operatorname{Sign}\left[1 - \frac{x}{2 t}\right]^2\right)$$

In[45]:=

```
Plot[{f4[x], u4[x, 0.005], u4[x, 0.05], u4[x, 0.5]},
  {x, -4, 4}, ImageSize → {500, 300}]
```

Out[45]=

In[46]:=

```
u5[x_, t_] := Evaluate[Integrate[Abs[f3[y]] s[x - y, t],
    {y, -Infinity, Infinity}, Assumptions → {t > 0}]]
```

In[47]:=

```
Simplify[u5[x, t]]
```

Out[47]=

$$\frac{1}{2} e^{t-x} \left(1 + \text{Erf}\left[\frac{-2 t + x}{2 \sqrt{t}}\right] + e^{2 x} \text{Erfc}\left[\frac{2 t + x}{2 \sqrt{t}}\right]\right)$$

In[48]:=

```
Plot[{Abs[f3[x]], u5[x, 0.005], u5[x, 0.05], u5[x, 0.5]},
 {x, -4, 4}, ImageSize → {500, 300}]
```

Out[48]=



In[49]:=

```
Show[%, PlotRange -> {{0, 4}, {0, 1}}]
```

Out[49]=